# NAG Library Routine Document

# G13DNF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

## 1 Purpose

G13DNF calculates the sample partial lag correlation matrices of a multivariate time series. A set of $\chi^2$-statistics and their significance levels are also returned. A call to G13DMF is usually made prior to calling this routine in order to calculate the sample cross-correlation matrices.

## 2 Specification

```
SUBROUTINE G13DNF (K, N, M, KMAX, R0, R, MAXLAG, PARLAG, X, PVALUE,        &
                   WORK, LWORK, IFAIL)
INTEGER            K, N, M, KMAX, MAXLAG, LWORK, IFAIL
REAL (KIND=nag_wp) R0(KMAX,K), R(KMAX,KMAX,M), PARLAG(KMAX,KMAX,M),        &
                   X(M), PVALUE(M), WORK(LWORK)
```

## 3 Description

Let $W_t = (w_{1t}, w_{2t}, \ldots, w_{kt})^{\mathrm{T}}$, for $t = 1, 2, \ldots, n$, denote $n$ observations of a vector of $k$ time series. The partial lag correlation matrix at lag $l$, $P(l)$, is defined to be the correlation matrix between $W_t$ and $W_{t+l}$, after removing the linear dependence on each of the intervening vectors $W_{t+1}, W_{t+2}, \ldots, W_{t+l-1}$. It is the correlation matrix between the residual vectors resulting from the regression of $W_{t+l}$ on the carriers $W_{t+l-1}, \ldots, W_{t+1}$ and the regression of $W_t$ on the same set of carriers; see Heyse and Wei (1985).

$P(l)$ has the following properties.

(i) If $W_t$ follows a vector autoregressive model of order $p$, then $P(l) = 0$ for $l > p$;

(ii) When $k = 1$, $P(l)$ reduces to the univariate partial autocorrelation at lag $l$;

(iii) Each element of $P(l)$ is a properly normalized correlation coefficient;

(iv) When $l = 1$, $P(l)$ is equal to the cross-correlation matrix at lag 1 (a natural property which also holds for the univariate partial autocorrelation function).

Sample estimates of the partial lag correlation matrices may be obtained using the recursive algorithm described in Wei (1990). They are calculated up to lag $m$, which is usually taken to be at most $n/4$. Only the sample cross-correlation matrices ($\hat{R}(l)$, for $l = 0, 1, \ldots, m$) and the standard deviations of the series are required as input to G13DNF. These may be computed by G13DMF. Under the hypothesis that $W_t$ follows an autoregressive model of order $s - 1$, the elements of the sample partial lag matrix $\hat{P}(s)$, denoted by $\hat{P}_{ij}(s)$, are asymptotically Normally distributed with mean zero and variance $1/n$. In addition the statistic

$$X(s) = n\sum_{i=1}^{k}\sum_{j=1}^{k}\hat{P}_{ij}(s)^2$$

has an asymptotic $\chi^2$-distribution with $k^2$ degrees of freedom. These quantities, $X(l)$, are useful as a diagnostic aid for determining whether the series follows an autoregressive model and, if so, of what order.

## 4 References

Heyse J F and Wei W W S (1985) The partial lag autocorrelation function *Technical Report No. 32* Department of Statistics, Temple University, Philadelphia

Wei W W S (1990) *Time Series Analysis: Univariate and Multivariate Methods* Addison–Wesley

## 5 Arguments

1: K – INTEGER *Input*

*On entry*: $k$, the dimension of the multivariate time series.

*Constraint*: $\mathrm{K} \geq 1$.

2: N – INTEGER *Input*

*On entry*: $n$, the number of observations in each series.

*Constraint*: $\mathrm{N} \geq 2$.

3: M – INTEGER *Input*

*On entry*: $m$, the number of partial lag correlation matrices to be computed. Note this also specifies the number of sample cross-correlation matrices that must be contained in the array R.

*Constraint*: $1 \leq \mathrm{M} < \mathrm{N}$.

4: KMAX – INTEGER *Input*

*On entry*: the first dimension of the arrays R0, R and PARLAG and the second dimension of the arrays R and PARLAG as declared in the (sub)program from which G13DNF is called.

*Constraint*: $\mathrm{KMAX} \geq \mathrm{K}$.

5: R0(KMAX, K) – REAL (KIND=nag_wp) array *Input*

*On entry*: if $i \neq j$, then $\mathrm{R0}(i,j)$ must contain the $(i,j)$th element of the sample cross-correlation matrix at lag zero, $\hat{R}_{ij}(0)$. If $i = j$, then $\mathrm{R0}(i,i)$ must contain the standard deviation of the $i$th series.

6: R(KMAX, KMAX, M) – REAL (KIND=nag_wp) array *Input*

*On entry*: $\mathrm{R}(i,j,l)$ must contain the $(i,j)$th element of the sample cross-correlation at lag $l$, $\hat{R}_{ij}(l)$, for $l = 1,2,\ldots,m$, $i = 1,2,\ldots,k$ and $j = 1,2,\ldots,k$, where series $j$ leads series $i$ (see Section 9).

7: MAXLAG – INTEGER *Output*

*On exit*: the maximum lag up to which partial lag correlation matrices (along with $\chi^2$-statistics and their significance levels) have been successfully computed. On a successful exit MAXLAG will equal M. If $\mathrm{IFAIL} = 2$ on exit, then MAXLAG will be less than M.

8: PARLAG(KMAX, KMAX, M) – REAL (KIND=nag_wp) array *Output*

*On exit*: $\mathrm{PARLAG}(i,j,l)$ contains the $(i,j)$th element of the sample partial lag correlation matrix at lag $l$, $\hat{P}_{ij}(l)$, for $l = 1,2,\ldots,\mathrm{MAXLAG}$, $i = 1,2,\ldots,k$ and $j = 1,2,\ldots,k$.

9: X(M) – REAL (KIND=nag_wp) array *Output*

*On exit*: $\mathrm{X}(l)$ contains the $\chi^2$-statistic at lag $l$, for $l = 1,2,\ldots,\mathrm{MAXLAG}$.

10: PVALUE(M) – REAL (KIND=nag_wp) array *Output*

*On exit*: PVALUE($l$) contains the significance level of the corresponding $\chi^2$-statistic in X, for $l = 1, 2, \ldots, \text{MAXLAG}$.

11: WORK(LWORK) – REAL (KIND=nag_wp) array *Workspace*

12: LWORK – INTEGER *Input*

*On entry*: the dimension of the array WORK as declared in the (sub)program from which G13DNF is called.

*Constraint*: $\text{LWORK} \geq (5\text{M} + 6)\text{K}^2 + \text{K}$.

13: IFAIL – INTEGER *Input/Output*

*On entry*: IFAIL must be set to 0, $-1$ or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value $-1$ or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value $-1$ or 1 is used it is essential to test the value of IFAIL on exit.**

*On exit*: $\text{IFAIL} = 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

# 6 Error Indicators and Warnings

If on entry $\text{IFAIL} = 0$ or $-1$, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

$\text{IFAIL} = 1$

On entry, $\text{K} < 1$,
or $\text{N} < 2$,
or $\text{M} < 1$,
or $\text{M} \geq \text{N}$,
or $\text{KMAX} < \text{K}$,
or $\text{LWORK} < (5\text{M} + 6)\text{K}^2 + \text{K}$.

$\text{IFAIL} = 2$

The recursive equations used to compute the sample partial lag correlation matrices have broken down at lag $\text{MAXLAG} + 1$. All output quantities in the arrays PARLAG, X and PVALUE up to and including lag MAXLAG will be correct.

$\text{IFAIL} = -99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

$\text{IFAIL} = -399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -999$

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

## 7 Accuracy

The accuracy will depend upon the accuracy of the sample cross-correlations.

## 8 Parallelism and Performance

G13DNF is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

G13DNF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

The time taken is roughly proportional to $m^2k^3$.

If you have calculated the sample cross-correlation matrices in the arrays R0 and R, without calling G13DMF, then care must be taken to ensure they are supplied as described in Section 5. In particular, for $l \geq 1$, $\hat{R}_{ij}(l)$ must contain the sample cross-correlation coefficient between $w_{i(t-l)}$ and $w_{jt}$.

The routine G13DBF computes squared partial autocorrelations for a specified number of lags. It may also be used to estimate a sequence of partial autoregression matrices at lags $1, 2, \ldots$ by making repeated calls to the routine with the argument NK set to $1, 2, \ldots$. The $(i, j)$th element of the sample partial autoregression matrix at lag $l$ is given by $W(i, j, l)$ when NK is set equal to $l$ on entry to G13DBF. Note that this is the 'Yule–Walker' estimate. Unlike the partial lag correlation matrices computed by G13DNF, when $W_t$ follows an autoregressive model of order $s - 1$, the elements of the sample partial autoregressive matrix at lag $s$ do not have variance $1/n$, making it very difficult to spot a possible cut-off point. The differences between these matrices are discussed further by Wei (1990).

Note that G13DBF takes the sample cross-covariance matrices as input whereas this routine requires the sample cross-correlation matrices to be input.

## 10 Example

This example computes the sample partial lag correlation matrices of two time series of length 48, up to lag 10. The matrices, their $\chi^2$-statistics and significance levels and a plot of symbols indicating which elements of the sample partial lag correlation matrices are significant are printed. Three * represent significance at the 0.5% level, two * represent significance at the 1% level and a single * represents significance at the 5% level. The * are plotted above or below the central line depending on whether the elements are significant in a positive or negative direction.

### 10.1 Program Text

```
!   G13DNF Example Program Text
!   Mark 26 Release. NAG Copyright 2016.

    Module g13dnfe_mod

!     G13DNF Example Program Module:
!            Parameters and User-defined Routines
```

```
!     .. Use Statements ..
      Use nag_library, Only: nag_wp
!     .. Implicit None Statement ..
      Implicit None
!     .. Accessibility Statements ..
      Private
      Public                           :: zprint
!     .. Parameters ..
      Integer, Parameter, Public       :: nin = 5, nout = 6
    Contains
      Subroutine zprint(k,n,m,ldpar,parlag,x,pvalue,nout)

!       .. Use Statements ..
        Use nag_library, Only: x04cbf
!       .. Scalar Arguments ..
        Integer, Intent (In)           :: k, ldpar, m, n, nout
!       .. Array Arguments ..
        Real (Kind=nag_wp), Intent (In) :: parlag(ldpar,ldpar,m), pvalue(m),   &
                                          x(m)
!       .. Local Scalars ..
        Real (Kind=nag_wp)             :: c1, c2, c3, c5, c6, c7, inv_sqrt_n,  &
                                          sum
        Integer                        :: i, i2, ifail, j, l, ll
!       .. Local Arrays ..
        Character (1)                  :: clabs(1), rlabs(1)
        Character (79)                 :: rec(7)
!       .. Intrinsic Procedures ..
        Intrinsic                      :: real, sqrt
!       .. Executable Statements ..
!       Print the partial lag correlation matrices.

        inv_sqrt_n = 1.0E0_nag_wp/sqrt(real(n,kind=nag_wp))
        Write (nout,*)
        Write (nout,*) ' PARTIAL LAG CORRELATION MATRICES'
        Write (nout,*) ' --------------------------------'
        Do l = 1, m
          Write (nout,99999) ' Lag = ', l
          Flush (nout)
          ifail = 0
          Call x04cbf('G','N',k,k,parlag(1,1,l),ldpar,'F9.3',' ','N',rlabs,    &
            'N',clabs,80,5,ifail)
        End Do
        Write (nout,99998) ' Standard error = 1 / SQRT(N) =', inv_sqrt_n

!       Print indicator symbols to indicate significant elements.
        Write (nout,*)
        Write (nout,*) ' TABLES OF INDICATOR SYMBOLS'
        Write (nout,*) ' ---------------------------'
        Write (nout,99999) ' For Lags 1 to ', m

!       Set up annotation for the plots.
        Write (rec(1),99997) '             0.005  :'
        Write (rec(2),99997) '       +     0.01   :'
        Write (rec(3),99997) '             0.05   :'
        Write (rec(4)(1:23),99997) '   Sig. Level        :'
        Write (rec(4)(24:),99997) '- - - - - - - - - -  Lags'
        Write (rec(5),99997) '             0.05   :'
        Write (rec(6),99997) '       -     0.01   :'
        Write (rec(7),99997) '             0.005  :'

!       Set up the critical values
        c1 = 3.29E0_nag_wp*inv_sqrt_n
        c2 = 2.58E0_nag_wp*inv_sqrt_n
        c3 = 1.96E0_nag_wp*inv_sqrt_n
        c5 = -c3
        c6 = -c2
        c7 = -c1

        Do i = 1, k
          Do j = 1, k
            Write (nout,*)
```

```
            If (i==j) Then
              Write (nout,99996) ' Auto-correlation function for', ' series ', &
                i
            Else
              Write (nout,99995) ' Cross-correlation function for',            &
                ' series ', i, ' and series', j
            End If
            Do l = 1, m
              ll = 23 + 2*l
              sum = parlag(i,j,l)

!             Clear the last plot with blanks
              Do i2 = 1, 7
                If (i2/=4) Then
                  rec(i2)(ll:ll) = ' '
                End If
              End Do

!             Check for significance
              If (sum>c1) Then
                rec(1)(ll:ll) = '*'
              End If
              If (sum>c2) Then
                rec(2)(ll:ll) = '*'
              End If
              If (sum>c3) Then
                rec(3)(ll:ll) = '*'
              End If
              If (sum<c5) Then
                rec(5)(ll:ll) = '*'
              End If
              If (sum<c6) Then
                rec(6)(ll:ll) = '*'
              End If
              If (sum<c7) Then
                rec(7)(ll:ll) = '*'
              End If
            End Do

!           Print
            Write (nout,99997)(rec(i2),i2=1,7)
          End Do
        End Do

!       Print the chi-square statistics and p-values.
        Write (nout,*)
        Write (nout,*)
        Write (nout,*) ' Lag     Chi-square statistic     P-value'
        Write (nout,*) ' ---     --------------------     -------'
        Write (nout,*)
        Write (nout,99994)(l,x(l),pvalue(l),l=1,m)

        Return

99999   Format (/,1X,A,I2)
99998   Format (/,1X,A,F6.3,A)
99997   Format (1X,A)
99996   Format (/,/,1X,A,A,I2,/)
99995   Format (/,/,1X,A,A,I2,A,I2,/)
99994   Format (1X,I4,10X,F8.3,11X,F8.4)
      End Subroutine zprint
    End Module g13dnfe_mod
    Program g13dnfe

!     G13DNF Example Main Program

!     .. Use Statements ..
      Use nag_library, Only: g13dmf, g13dnf, nag_wp
      Use g13dnfe_mod, Only: nin, nout, zprint
!     .. Implicit None Statement ..
      Implicit None
```

```
!     .. Local Scalars ..
      Integer                          :: i, ifail, k, kmax, lwork, m, maxlag, &
                                          n
      Character (1)                    :: matrix
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: parlag(:,:,:), pvalue(:), r(:,:,:),  &
                                          r0(:,:), w(:,:), wmean(:), work(:),  &
                                          x(:)
!     .. Executable Statements ..
      Write (nout,*) 'G13DNF Example Program Results'
      Write (nout,*)

!     Skip heading in data file
      Read (nin,*)

!     Read in the problem size
      Read (nin,*) k, n, m, matrix

      kmax = k
      lwork = (5*m+6)*k**2 + k
      Allocate (w(kmax,n),wmean(k),r0(kmax,k),r(kmax,kmax,m),                  &
        parlag(kmax,kmax,m),x(m),pvalue(m),work(lwork))

!     Read in series
      Read (nin,*)(w(i,1:n),i=1,k)

!     Calculate cross correlations
      ifail = 0
      Call g13dmf(matrix,k,n,m,w,kmax,wmean,r0,r,ifail)

!     Calculate sample partial lag correlation matrices
      ifail = 0
      Call g13dnf(k,n,m,kmax,r0,r,maxlag,parlag,x,pvalue,work,lwork,ifail)

!     Display results
      Call zprint(k,n,m,kmax,parlag,x,pvalue,nout)

    End Program g13dnfe
```

## 10.2 Program Data

```
G13DNF Example Program Data
2 48 10 'R'                                                :: K,N,M,MATRIX
-1.490 -1.620  5.200  6.230  6.210  5.860  4.090  3.180
 2.620  1.490  1.170  0.850 -0.350  0.240  2.440  2.580
 2.040  0.400  2.260  3.340  5.090  5.000  4.780  4.110
 3.450  1.650  1.290  4.090  6.320  7.500  3.890  1.580
 5.210  5.250  4.930  7.380  5.870  5.810  9.680  9.070
 7.290  7.840  7.550  7.320  7.970  7.760  7.000  8.350
 7.340  6.350  6.960  8.540  6.620  4.970  4.550  4.810
 4.750  4.760 10.880 10.010 11.620 10.360  6.400  6.240
 7.930  4.040  3.730  5.600  5.350  6.810  8.270  7.680
 6.650  6.080 10.250  9.140 17.750 13.300  9.630  6.800
 4.080  5.060  4.940  6.650  7.940 10.760 11.890  5.850
 9.010  7.500 10.020 10.380  8.150  8.370 10.730 12.145 :: End of W
```

## 10.3 Program Results

```
 G13DNF Example Program Results


  PARTIAL LAG CORRELATION MATRICES
  --------------------------------

  Lag =  1
           0.736    0.174
           0.211    0.555

  Lag =  2
          -0.187   -0.083
```

```
        -0.180   -0.072

Lag =  3
         0.278   -0.007
         0.084   -0.213

Lag =  4
        -0.084    0.227
         0.128   -0.176

Lag =  5
         0.236    0.238
        -0.047   -0.046

Lag =  6
        -0.016    0.087
         0.100   -0.081

Lag =  7
        -0.036    0.261
         0.126    0.012

Lag =  8
         0.077    0.381
         0.027   -0.149

Lag =  9
        -0.065   -0.387
         0.189    0.057

Lag = 10
        -0.026   -0.286
         0.028   -0.173

Standard error = 1 / SQRT(N) = 0.144

TABLES OF INDICATOR SYMBOLS
---------------------------

For Lags 1 to 10


Auto-correlation function for series  1

              0.005  : *
        +     0.01   : *
              0.05   : *
   Sig. Level        : - - - - - - - - - -  Lags
              0.05   :
        -     0.01   :
              0.005  :


Cross-correlation function for series  1 and series 2

              0.005  :
        +     0.01   :               *
              0.05   :               *
   Sig. Level        : - - - - - - - - - -  Lags
              0.05   :                 * *
        -     0.01   :                 *
              0.005  :


Cross-correlation function for series  2 and series 1

              0.005  :
        +     0.01   :
```

```
           0.05   :
   Sig. Level     : - - - - - - - - - - -  Lags
           0.05   :
      -    0.01   :
           0.005  :


Auto-correlation function for series  2

           0.005  : *
      +    0.01   : *
           0.05   : *
   Sig. Level     : - - - - - - - - - - -  Lags
           0.05   :
      -    0.01   :
           0.005  :

Lag     Chi-square statistic     P-value
---     --------------------     -------

  1           44.362              0.0000
  2            3.824              0.4304
  3            6.219              0.1834
  4            5.094              0.2778
  5            5.609              0.2303
  6            1.170              0.8830
  7            4.098              0.3929
  8            8.371              0.0789
  9            9.244              0.0553
 10            5.435              0.2455
```